



combien paieriez-vous d'impôts en 1980 ?

Questions que pose le programme

- Nombre de parts
- Nombre d'adultes
- Nombre d'enfants
- Nombre de salaires

Avez-vous des revenus mobiliers ?

NON

OUI

- revenus des valeurs et capitaux
- avoir fiscal
- abattement forfaitaire ou Monory

Forfaitaire

(déduction de 3 000 F au plus)

Monory

(déduction de 5 000 F + 500 F pour les 2 premiers enfants + 1 000 F par enfant suivant)

Par salaire

- Total des salaires
- Frais réels ou non

NON

(abattement de 10 %)

OUI

— Montant des frais réels

— Déduction supplémentaire ou non

(Si oui pourcentage de déduction, le montant en est limité à 40 000 F)
(calcul automatique de l'abattement de 20 % limité à 72 000 F)

Frais de garde (limités à 3 000 F par enfant)

— **Avez-vous des charges déductibles ?**

NON

OUI

- intérêts des emprunts
- dépenses de ravalement
- dépenses économies d'énergie
(le total de ces trois données est limité à 7 000 F + 1 000 F par enfant à charge avec un maximum de 10 000 F)
- pensions
- virement aux œuvres (limité à 1,5 % du revenu net)
- primes d'assurances vie
- abattement supplémentaire

Editions

Le programme donne

- le revenu net global imposable
- le montant de l'impôt
- et le cas échéant le montant de l'impôt après imputation de l'avoir fiscal.

Ce n'est pas une question indiscreète de notre part, mais, hélas, une triste réalité que l'on doit affronter tous les ans.

Cet affrontement serein amène parfois certaines surprises *a posteriori* (« Un troisième tiers provisionnel de ce montant ? Comment est-ce possible ? »), ou certaines interrogations *a priori* (« Si je souscris 4 500 francs d'actions au titre de la loi Monory, quelle sera mon économie d'impôt ? »). Sans parler du fait que calculer ses impôts en tenant compte des divers textes législatifs en vigueur à un instant donné relève parfois de l'acrobatie mathématique la plus pure.

La complexité de la procédure à suivre a évidemment amené l'Administration Fiscale à calculer les impôts des Français à partir des données brutes que ceux-ci veulent bien (?) lui fournir. Le temps où les Français devaient calculer à la main le montant de leurs impôts et le reporter sur leur déclaration est révolu. Aujourd'hui ils précisent leurs revenus divers et les déductions auxquelles ils aspirent, l'Administration fait le reste (!). Libre au contribuable de déterminer ensuite le montant de son impôt.

Le programme dont le listing est présenté ci-contre vous permettra d'obtenir automatiquement le montant de vos impôts pour l'année 1980. Il a été écrit en Basic pour un TRS-80 mais ne comporte pratiquement pas d'ordres spécifiques au TRS-80 et devrait consécutivement, être transposable sur tout ordinateur individuel « parlant » Basic.

Données nécessaires au fonctionnement

Le programme ne sollicite de l'utilisateur qu'un certain nombre de

```

10 REM PROGRAMME DE CALCUL DE L'IMPOT SUR LE REVENU
20 REM AUTEUR JEAN-PIERRE BRUNERIE
30 REM COPYRIGHT L'ORDINATEUR INDIVIDUEL ET L'AUTEUR
40 REM *****
50 REM VARIABLES
100 REM I1 : NOMBRE DE PARTS
110 REM I2 : NOMBRE D'ADULTES
120 REM I3 : NOMBRE D'ENFANTS
130 REM I4 : NOMBRE DE SALAIRES (MAXIMUM 4)
140 REM I5 : REVENUS DES CAPITAUX ET VALEURS MOBILIERES
150 REM I6 : AVOIR FISCAL
160 REM I7 : ACHATS MONORY
170 REM I8(I4) : TOTAL DES SALAIRES
180 REM I9(I4) : MONTANT DES FRAIS REELS OU DEDUCTION 10%
190 REM A1$ : FRAIS REELS OU NON
200 REM J9 : CALCUL DU BUTOIR DES INTERETS DEDUCTIBLES
210 REM A2$ : DEDUCTION SUPPLEMENTAIRE OU NON
220 REM K2 : T DE DEDUCTION SUPPLEMENTAIRE
230 REM A6$ : FORFAITAIRE OU MONORY
240 REM A4$ : AVOIR FISCAL OU NON
250 REM A5$ : TOP RECOMMENCEZ-VOUS OU NON
260 REM J1 : FRAIS DE GARDE
270 REM J2 : INTERETS DEDUCTIBLES DES EMPRUNTS
280 REM J3 : DEPENSES DE RAVALEMENT
290 REM J4 : DEPENSES ECONOMIES D'ENERGIE
300 REM J5 : PENSIONS
310 REM J6 : VERSEMENT AUX OEUVRES
320 REM J7 C : PRIMES ASSURANCE-VIE
330 REM J8 : ABATTEMENT SUPPLEMENTAIRE
340 REM K3(I4) : SALAIRES GLOBAUX
350 REM K2 : REVENUS PROFESSIONNELS NETS
360 REM K1 : REVENU NET GLOBAL IMPOSABLE
370 REM L1 : MONTANT DE L'IMPOT (CALCUL SIMPLE)
380 REM L2 : MONTANT DE L'IMPOT (APRES AVOIR FISCAL)
390 REM M1 : NOMBRE DE TRANCHES
400 REM M2(M1) : LIMITES DE TRANCHES
410 REM M3(M1) : TAUX D'IMPOSITION DE LA TRANCHE
420 REM *****
430 REM MODE D'EMPLOI
440 CLEAR
450 PRINT "CE PROGRAMME VOUS DONNERA LE MONTANT DE L'IMPOT"
460 PRINT "DONT VOUS ETES REDEVABLE AU TRESOR FRANCAIS POUR"
470 PRINT "VOS REVENUS DE L'ANNEE 1979."
480 PRINT "IL S'APPUIE SUR LES DONNEES QUE VOUS VOUDREZ"
490 PRINT "BEN LUI FOURNIR A PARTIR DE LA PAGE 4 DE LA "
500 PRINT "NOTICE D'EXPLICATION DE LA DECLARATION DES"
510 PRINT "REVENUS DE L'ANNEE 1979."
520 PRINT "IL EFFECTUE L'ENSEMBLE DES CALCULS INTERMEDIAIRES"
530 PRINT "ET VOUS DONNERA LE MONTANT DE L'IMPOT (AVEC LE"
540 PRINT "CAS ECHEANT, LE MONTANT AVEC AVOIR FISCAL)"
550 PRINT "BON COURAGE!!!!!"
560 INPUT "ETES VOUS PRET (O-N) ? ";A5$
570 IF A5$="O" THEN 600
580 PRINT "TANT PIS ...."
590 REM *****
600 REM INITIALISATION
610 DATA 0
620 DATA .00
630 DATA 8725
640 DATA .05
650 DATA 9125
660 DATA .10
670 DATA 10825
680 DATA .15
690 DATA 17125
700 DATA .20
710 DATA 22275
720 DATA .25
730 DATA 28000
740 DATA .30
750 DATA 33875
760 DATA .35
770 DATA 39075
780 DATA .40
790 DATA 65125
800 DATA .45
810 DATA 89575
820 DATA .50
830 DATA 105950
840 DATA .55
850 DATA 125050
860 DATA .60
870 DIM I8(4),I9(4),K3(4),M2(14),M3(13)
880 M2(14)=9999999
890 M1=13
900 FOR I=1 TO M1
910 READ M2(I)
920 READ M3(I)
930 NEXT I
940 REM *****
1000 REM *****
1010 REM SAISIE
1020 CLS
1030 INPUT "NOMBRE DE PARTS : ";I1
1040 INPUT "NOMBRE D'ADULTES : ";I2
1050 INPUT "NOMBRE D'ENFANTS : ";I3
1060 INPUT "NOMBRE DE SALAIRES : ";I4
1070 IF I4>4 THEN PRINT "(IMPERIEUR OU EGAL A 5)"; GOTO 1060
1080 REM *****

```

```

1090 PRINT
1100 INPUT "AVEZ-VOUS DES REVENUS MOBILIERES (O-N)";A4$
1110 IF A4$="O" THEN 1170
1120 IF A4$<"N" THEN 1100
1130 I5=0
1140 I6=0
1150 I7=0
1160 GOTO 2000
1170 INPUT "REVENUS DES VALEURS ET CAPITAUX : ";I5
1180 INPUT "AVOIR FISCAL : ";I6
1190 INPUT "ABATTEMENT FORFAITAIRE OU MONORY (F-N)";A6$
1200 IF A6$="F" THEN 1260
1210 IF A6$<"N" THEN 1190
1220 INPUT "ACHATS MONORY (SOLDE NET) : ";I7
1230 IF I3>2 THEN J9=6000+1000*(I3-2) ELSE J9=5000+500*I3
1240 IF I7>J9 THEN I7=J9; PRINT "** ABATTEMENT RETENU ";I7
1250 GOTO 2000
1260 PRINT "VOTRE REVENU DOIT ETRE GLOBALEMENT INFERIEUR"
1270 PRINT "A 180.000 FRANCS"; IF I5<3000 THEN I7=I5 ELSE I7=3000
1280 PRINT "** ABATTEMENT RETENU ";I7
2000 REM SAISIE DES SALAIRES
2010 FOR I=1 TO I4
2020 PRINT
2030 PRINT "TOTAL DES SALAIRES (";I;")";INPUT I8(I)
2040 PRINT "FRAIS REELS (O-N) : ";INPUT A1$
2050 IF A1$="N" THEN XX=INT(I8(I)*.10);
IF XX<1800 THEN I9(I)=1800; GOTO 2080
ELSE I9(I)=XX : GOTO 2080
2060 IF A1$<"O" THEN 2040
2070 PRINT "MONTANT DES FRAIS REELS (";I;")";INPUT I9(I); GOTO 2170
2080 PRINT "** DEDUCTION POUR FRAIS ";I9(I);
PRINT "DEDUCTION SUPPLEMENTAIRE (O-N) ";INPUT A2$
2090 IF A2$="N" THEN 2170
2100 IF A2$<"O" THEN 2080
2110 PRINT "POURCENTAGE DE DEDUCTION : ";INPUT A3
2120 A3=A3/100
2130 J9=INT((I8(I)-I9(I))*A3)
2140 IF J9>40000 THEN J9=40000
2150 PRINT "** DEDUCTION RETENUE ";J9
2160 I9(I)=I9(I)+J9
2170 J9=INT((I8(I)-I9(I))*2)
2180 IF J9>72000 THEN J9=72000
2190 PRINT "** DEDUCTION POUR SALAIRE ";J9
2200 K3(I)=(I8(I)-I9(I))-J9
2210 K2=K2+K3(I)
2220 PRINT "** TOTAL IMPOSABLE POUR CE SALAIRE ";K3(I)
2230 PRINT "*** CUMUL IMPOSABLE DES SALAIRES ";K2
2240 NEXT I
3000 PRINT
3010 INPUT "FRAIS DE GARDE : ";J1
3020 J9=3000+J1
3030 IF J1>J9 THEN J1=J9 : PRINT "** DEDUCTION RETENUE ";J1
3040 K2=K2+J1
3050 IF K2<J1 THEN K2=J1
3060 K2=K2-J1
3070 PRINT
3080 INPUT "AVEZ-VOUS DES CHARGES DEDUCTIBLES ? ";A5$
3090 IF A5$="N" THEN K1=K2-17 : GOTO 3250
3100 IF A5$<"O" THEN 3080
3110 PRINT "CHARGES DEDUCTIBLES"
3120 INPUT "-INTERETS DES EMPRUNTS : ";J2
3130 INPUT "-DEPENSES DE RAVALEMENT : ";J3
3140 INPUT "-DEPENSES ECONOMIES D'ENERGIE : ";J4
3150 J9=7000+13*1000; IF J9>10000 THEN J9=10000
3160 J2=J2+J3+J4
3170 IF J2>J9 THEN J2=J9; PRINT "** TOTAL RETENU ";J2
3180 INPUT "-PENSIONS : ";J5
3190 INPUT "-VIREMENT AUX OEUVRES : ";J6
3200 IF J6*(K2*.015) THEN J6=INT(K2*.015) : PRINT "** MONTANT RETENU ";J6
3210 INPUT "-PRIMES D'ASSURANCE-VIE : ";J7
3220 INPUT "-ABATTEMENT SUPPLEMENTAIRE : ";J8
3230 K1=K2-J2-J5-J6-J7-J8-17
3240 PRINT
3250 PRINT "** REVENU NET GLOBAL IMPOSABLE";K1
4000 REM *****
4010 REM CALCUL DU MONTANT DE L'IMPOT
4020 K1=INT(K1/100+.001)*100;K1=INT(K1/I1)
4030 FOR I=1 TO M1
4040 KA=K1-M2(I)
4050 IF KA>(M2(I+1)-M2(I)) THEN L1=L1+INT((M2(I+1)-M2(I))*M3(I))
ELSE L1=L1+INT(KA*M3(I)) : I=M1
4060 NEXT I
4070 L1=INT(L1*I1)
4080 IF A4$="N" THEN L2=L1 ELSE L2=L1-I6
5000 REM *****
5010 REM EDITION
5020 PRINT
5030 PRINT "VOTRE IMPOT SE MONTERA A ";L1;" FRANCS"
5040 IF A4$="N" THEN 5070
5050 PRINT "ET APRES DEDUCTION DE L'AVOIR FISCAL "
5060 PRINT "A ";L2;" FRANCS"
5070 PRINT "J'ESPERE QUE CE N'ETAIT PAS UNE TROP "
5080 PRINT "MAUVAISE NOUVELLE."
5090 INPUT "VOULEZ-VOUS RECOMMENCER ? ";A5$
5100 IF A5$="O" THEN 440
5110 IF A5$="N" THEN 5140
5120 PRINT "PARDON ?"
5130 GOTO 5090
5140 END

```

données qui sont toutes comprises dans votre déclaration de revenus.

Elles correspondent globalement aux informations que l'on peut reporter sur la page 4 de la « Notice pour remplir votre déclaration des revenus de 1979 ».

Les seules informations nécessaires en sus sont :

- le nombre de parts
- le nombre d'adultes et le nombre d'enfants
- le nombre de salaires (4 maximum)

- le montant de l'avoir fiscal
- votre choix entre abattement forfaitaire ou Monory.

Limites

Ce programme convient à des personnes salariées ou n'ayant que des revenus immobiliers. Il ne concerne pas en particulier les professions non salariées ou les gérants de société, mais les adaptations dans ces deux cas sont simples et sont à effectuer dans la partie « saisie ».

Exemple d'exécution

Soit un ménage ayant 3 enfants. Il n'y a qu'un seul salaire de 70 000 F ; les seules déductions applicables sont 10 % et 20 %. Il n'y a pas de frais de garde ni de charges déductibles. La famille a souscrit pour 10 000 F d'actions qui ont rapporté 1 000 F de dividendes assortis d'un avoir fiscal de 500 F. L'impôt résultant est de 1 641 francs et de 1 141 francs après avoir fiscal.

Jean-Pierre Brunerie